

[Description](#)
[Options](#)[Quick start](#)
[Remarks and examples](#)[Menu](#)
[Also see](#)[Syntax](#)

Description

`twoway mspline` calculates cross medians and then uses the cross medians as knots to fit a cubic spline. The resulting spline is graphed as a line plot.

Quick start

Median-spline plot of y versus x

```
twoway mspline y x
```

Same as above, and overlay the median-spline plot on a scatterplot of the observed data

```
twoway scatter y x || mspline y x
```

Same as above, but specify 10 cross-median knots

```
twoway scatter y x || mspline y x, bands(10)
```

Specify 5 points between knots

```
twoway scatter y x || mspline y x, n(5)
```

Separate graph areas for each level of `catvar`

```
twoway scatter y x || mspline y x, n(5) by(catvar)
```

Label the y axis from 0 to 10 in increments of 2

```
twoway scatter y x || mspline y x, n(5) ylabel(0(2)10)
```

Specify “My X Variable” as the title for the x axis

```
twoway scatter y x || mspline y x, xtitle("My X Variable")
```

Same as above, but suppress the legend

```
twoway scatter y x || mspline y x, xtitle("My X Variable") ///  
legend(off)
```

Menu

Graphics > Two-way graph (scatter, line, etc.)

Syntax

```
twoway mspline yvar xvar [if] [in] [ , options]
```

<i>options</i>	Description
<code>bands(#)</code>	number of cross-median knots
<code>n(#)</code>	number of points between knots
<i>cline_options</i>	change look of the line
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All options are *rightmost*; see [G-4] **Concept: repeated options**.

Options

`bands(#)` specifies the number of bands for which cross medians should be calculated. The default is $\max\{\min(b_1, b_2), b_3\}$, where b_1 is $\text{round}\{10 * \log_{10}(N)\}$, b_2 is $\text{round}(\sqrt{N})$, b_3 is $\min(2, N)$, and N is the number of observations.

The x axis is divided into $\#$ equal-width intervals and then the median of y and the median of x are calculated in each interval. It is these cross medians to which a cubic spline is then fit.

`n(#)` specifies the number of points between the knots for which the cubic spline should be evaluated. `n(10)` is the default. `n()` does not affect the result that is calculated, but it does affect how smooth the result appears.

cline_options specify how the median-spline line is rendered and its appearance; see [G-3] *cline_options*.

axis_choice_options associate the plot with a particular y or x axis on the graph; see [G-3] *axis_choice_options*.

twoway_options are a set of common options supported by all *twoway* graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] *twoway_options*.

Remarks and examples

Remarks are presented under the following headings:

Typical use


Cautions

Use with by()

Typical use

Median splines provide a convenient way to show the relationship between y and x :

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)
. scatter mpg weight, msize(*.5) || mspline mpg weight
```




The important part of the above command is `mspline mpg weight`. On the scatter, we specified `msize(*.5)` to make the marker symbols half their normal size; see [\[G-4\] size](#).

Cautions

The graph shown above illustrates a common problem with this technique: it tracks wiggles that may not be real and can introduce wiggles if too many bands are chosen. An improved version of the graph above would be


```
. scatter mpg weight, msize(*.5) || mspline mpg weight, bands(8)
```



Use with by()

mspline may be used with by() (as can all the twoway plot commands):

```
. scatter mpg weight, msize(*.5) ||  
  mspline mpg weight, bands(8) ||, by(foreign, total row(1))
```



Also see

[R] [makespline](#) — Spline generation

[G-2] [graph twoway fpfit](#) — Two-way fractional-polynomial prediction plots

[G-2] [graph twoway lfit](#) — Two-way linear prediction plots

[G-2] [graph twoway line](#) — Two-way line plots

[G-2] [graph twoway mband](#) — Two-way median-band plots

[G-2] [graph twoway qfit](#) — Two-way quadratic prediction plots

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

